
1 Introduction 

Due to the even growing complexity of metamodels, the establishment of metamodel metrics now 
requires the use of dedicated tools. This ATL transformation example aims to show how the ATL 
language may be used to meet this objective by generating metrics models from metamodels. 

The Kernel MetaMetaModel (KM3) is a simple textual notation enabling quick definition of metamodels 
[1]. The KM3 to Metrics ATL example aims to demonstrate the feasibility of computing some 
metamodel metrics by means of the ATL language. For this purpose, we have developed a basic 
Metrics metamodel enabling to encode measures of different types. 

The KM3 to Metrics ATL transformation defines different metamodel metrics that are relative either to 
the whole model, or to specific model elements. The metrics provided in the scope of this 
transformation include simple measurements as well as more complex computations. 

2 The KM3 to Metrics ATL transformation 

2.1 Transformation overview 

The KM3 to Metrics transformation is a single step transformation that produces a Metrics model from 
a KM3 model. 

Users of the ATL Development Tools (ADT) [2] can easily produce their own KM3 input model by 1) 
entering a textual KM3 metamodel and, 2) exporting the produced textual KM3 metamodel into a KM3 
model by means of the Inject KM3 file to KM3 model contextual menu option. 

2.2 Metamodels 

The KM3 to Metrics transformation is based on both the KM3 and Metrics metamodel. The KM3 
descriptions of these metamodels can respectively be found in Appendix A: and Appendix B:. They are 
further described in the following subsections. 

2.2.1 The KM3 metamodel 

The KM3 metamodel [1] provides semantics for metamodel descriptions. The KM3 metamodel 
conforms to itself and can therefore be used to define KM3 metamodels. Figure 1 provides a 
description of a subset of the KM3 metamodel. Its corresponding complete textual description in the 
KM3 format is also provided in Appendix A:. 

A KM3 Metamodel is composed of Packages. A Package contains some abstract ModelElements 
(TypedElements, Classifiers, EnumLiterals and Packages, since a Package is itself a ModelElement). 
A ModelElement is characterized by its name. The ModelElement entity inherits from the abstract 
LocatedElement entity. This last defines a location attribute that aims to encode, in a string format, the 
location of the declaration of the corresponding element within its source file. 

A Classifier can be either an Enumeration, a DataType or a Class. An Enumeration is composed of 
EnumLiteral elements. The Class element defines the Boolean isAbstract attribute that enables to 
declare abstract classes. A Class can have direct supertypes (Class elements). 

A Class is composed of abstract StructuralFeatures. The StructuralFeature element inherits from the 
abstract TypedElement entity. This entity defines the lower, upper, isOrdered and isUnique attributes. 
The two first attributes defines the minimal and maximal cardinality of a TypedElement. The isOrdered 
and isUnique Boolean attributes respectively encode the fact that the different instances of the 

TypedElement are ordered and unique. A TypedElement obviously has a type, which corresponds to a 
Classifier element. 

 
Figure 1. The KM3 metamodel 

A StructuralFeature is either a Reference or an Attribute. The Reference element defines the Boolean 
isContainer attribute that encode the fact that the pointed elements are contained by the reference. A 
Reference can also have an opposite reference. Finally, a StructuralFeature has an owner of the type 
Class (the owner reference is the opposite of the Class structuralFeatures reference). 

2.2.2 The Metrics metamodel 

The Metrics metamodel is a simple metamodel for metrics encoding. Figure 2 provides a description of 
this metamodel. The main element of the Metrics metamodel is the abstract Metric entity. This entity 
defines two string attributes that are common to the different types of encoded metrics: the context 
and the label of the Metric. The context aims to identify, by means of a string value, the KM3 element 
the metric relies to. The label, as for it, provides a textual description of the metric (e.g. what is 
measured). 

Metric

+context : string

+label : string

+data : int

IntegerMetric

+data : bool

BooleanMetric

+data : string

StringMetric

 
Figure 2. The Metrics metamodel 

The Metrics metamodel defines three types of concrete metrics, which all inherit from the abstract 
Metric entity: the IntegerMetric, the BooleanMetric and the StringMetric. They respectively encode 
integral, boolean and string measurements. Each of them has a data attribute, of the type that 
corresponds to the metric type, which encodes the metric value. 

2.3 Rules specification 

Here are the rules used to generate a Metrics model from a KM3 model: 

• Five IntegerMetric elements, corresponding to the number of classes and associations of the 
input KM3 model, as well as its maximal inheritance depth and its number of inheritance 
trees and graphs, are generated from a KM3 Metamodel; 

• Four IntegerMetrics elements, corresponding to the number of own attributes and own 
references of the input Class element, as well as the total number (including inherited ones) 
of attributes and references of the Class, are generated from a KM3 Class. 

The maximal inheritance depth corresponds to the maximal length of the inheritance chains of the 
input model. 

An independent inheritance tree corresponds to a set of classes of the input model that have the same 
root class. A class without any supertype forms a single element inheritance tree. Note that a given 
class may belong to different inheritance trees if it has multiple inheritances. 

An independent inheritance graph corresponds to a set of classes of the input model that are linked to 
each other by means of an inheritance relationship (either supertype or subtype). A class with no 
supertypes nor subtypes forms a single element inheritance graph. Note that, unlike the with an 
inheritance tree, a given class belongs to a single inheritance graph. As a consequence, the number 
of inheritance graphs associated with a given model is lower or equal to its number of inheritance 
trees. 

2.4 ATL code 

The ATL code for the UML to MOF transformation is provided in Appendix C:. It consists of 15 helpers 
and 2 rules. 

2.4.1 Helpers 

The allClasses helper computes the set of all Class elements of the input KM3 model. Since this set 
is used several times, defining this constant helper enables to avoid multiple calculations of the set of 
classes. 

The allReferences helper computes the set of all Reference elements of the input KM3 model. Since 
this set is used several times, defining this constant helper enables to avoid multiple calculations of the 
set of references. 

The inheritanceRoots helper computes the set of all Class elements that correspond to the root of an 
inheritance tree. These Class elements are those that have no supertypes. Since this set is used 
several times, defining this constant helper enables to avoid multiple calculations of the set of classes 

The attributeNb helper computes the number of own attributes of its contextual Class. As a constant 
helper, it calculates this value only once (at the time it is called for the first time) and stores it as a 
cached value. 

The referenceNb helper is similar to the attributeNb helper, except that it deals with the references of 
its contextual Class. 

The attributeNb2 helper computes the total number of attributes (including inherited ones) of its 
contextual Class. This helper is a recursive helper which returns the number of attributes of its 
contextual Class plus the values returned by its recursive calls on each supertype of this contextual 
Class. As a constant helper, it calculates this value only once (at the time it is called for the first time) 
and stores it as a cached value. 

The referenceNb2 helper is similar to the attributeNb2 helper, except that it deals with the references 
of its contextual Class. 

The getInheritanceLenght() helper aims to compute the maximal length of the inheritance chains that 
lead to the contextual Class. This helper is a recursive helper. If the contextual Class has no 
supertypes, it returns 0. Otherwise, it returns the maximal value computed by its different recursive 
calls on the supertypes of the contextual Class (the length of the inheritance chain is incremented by 1 
at each successive recursive call). 

The getInheritanceMaxDepth() helper aims to compute the maximal length of an inheritance chain 
within the input KM3 model. For this purpose, it iterates the classes of the input model, calculating for 
their own inheritance length by calling the getInheritanceLenght helper. It then returns the maximal 
value computed by this mean. 

The getInheritanceTreeNb() helper calculates the number of inheritance trees in the input model. 
This number corresponds to the number of Class elements that are roots of these trees. As a 
consequence, the helper computes the number of tree root Class elements, that is the number of 
Class that has no supertypes. 

The subtypes helper aims to compute the set of direct subtypes of the contextual Class. For this 
purpose, it iterates the Class elements of the input model, and selects those that have the contextual 
Class as a supertype. As a constant helper, it calculates this value only once (at the time it is called for 
the first time) and stores it as a cached value. 

The getTree() helper calculates the set of Class elements composing the inheritance tree that has the 
contextual Class as root element. This helper is a recursive helper. If the contextual Class has no 
supertypes, it returns a set containing the only contextual Class. Otherwise, the helper iterates the 
supertypes of the contextual Class and returns a set containing this contextual Class along with the 
results of its recursive calls on the supertypes of the Class. 

The existsHLink() helper returns a boolean value stating whether there exists any inheritance 
relationships between the Class elements of the two set of classes it receives as parameters. For this 
purpose, it iterates the Class elements of the first set and checks whether there exists an inheritance 
link (either a supertype or a subtype relation) between this current Class and any Class of the second 
set. The helper returns true if it finds out such an inheritance relation, false otherwise. 

The computeGraphNb() helper aims to compute the number of inheritance graphs corresponding to 
the sequence of inheritance trees passed as a parameter. This helper is a recursive helper. If the 
provided sequence contains a single tree, the helper returns 1, otherwise, it calculates a new 

sequence of inheritance trees by trying to merge the first tree of the sequence with the following ones 
according to the inheritance links that may exist between the classes of these trees. Indeed, two 
inheritance trees of a same model belong to the same inheritance graph if there exists at least an 
inheritance relation between one Class of each tree. The helper therefore iterates the input tree 
sequence and checks whether it exists an inheritance relation between the first tree and the currently 
iterated one (by calling existsHLink the helper). If so, both trees are merged in order to form the new 
first tree of the new sequence. If the newly computed sequence contains a single tree, the helper 
returns 1. Otherwise, the helper returns 1 plus the value returned by its recursive call on a sequence 
that corresponds to the newly computed sequence without its first tree. Indeed, after the successive 
mergings, this last corresponds to an inheritance graph of the input model and has no inheritance links 
with the remaining trees. 

The getInheritanceGraphNb() helper aims to compute the number of inheritance graphs of the input 
KM3 model. To this end, its first computes a sequence containing the inheritance trees of the input 
model. This is simply achieved by iterating the tree root Class elements and calling the getTree helper 
on each of them. The helper then returns the value returned by the computeGraphNb helper called 
with the calculated sequence as a parameter. 

2.4.2 Rules 

The Model rule generates Metrics elements that are relative to the input KM3 model. This rule 
generates 5 IntegerMetric elements. The first one corresponds to the number of Class of the input 
KM3 model. The second one is associated with the number of associations in the model. This value 
corresponds to the number of references without an opposite, plus the half of the number of 
references that have an opposite (since two opposite references represent a same association). The 
third generated metric corresponds to the maximal inheritance length in the input model, and is 
computed by the getInheritanceMaxDepth() helper. The two last metrics correspond to the number of 
inheritance trees and graphs within the KM3 model. They are respectively computed by the 
getInheritanceTreeNb() and getInheritanceGraphNb() helpers. 

The Class rule generates Metrics elements that are relative to Class elements of the input KM3 
model. This rule generates 4 IntegerMetric elements: one devoted to the number of own attributes of 
the current class, one to its number of own references, one to its total number of attribute and one to 
its total number of references. The values of the data attribute of the generated Metrics are 
respectively provided by the attributeNb, referenceNb, attributeNb2 and referenceNb2 helpers. 

3 References 

[1] KM3 User Manual. The Eclipse Generative Model Transformer (GMT) project, http://eclipse.org/gmt/. 

[2] The ATL Development Tools (ADT). The Eclipse Generative Model Transformer (GMT) project, 

http://eclipse.org/gmt/. 

Appendix A: The KM3 metamodel in KM3 
format 

package KM3 { 1 
 2 
 abstract class LocatedElement { 3 
  attribute location : String; 4 
  attribute commentsBefore [*] ordered : String; 5 
  attribute commentsAfter [*] ordered : String; 6 
 } 7 
 8 
 abstract class ModelElement extends LocatedElement { 9 
  attribute name : String; 10 
  reference "package"  : Package oppositeOf contents ; 11 
 } 12 
 13 
 class Classifier extends ModelElement { 14 
 } 15 
 16 
 class DataType extends Classifier { 17 
 } 18 
 19 
 class Enumeration extends Classifier { 20 
  reference literals [*] ordered container : EnumLiteral oppositeOf enum ; 21 
 } 22 
 23 
 class EnumLiteral extends ModelElement { 24 
  reference enum : Enumeration oppositeOf literals ; 25 
 } 26 
 27 
 class TemplateParameter extends Classifier { 28 
 } 29 
 30 
 class Class extends Classifier { 31 
  reference parameters [*] ordered container : TemplateParameter ; 32 
  attribute isAbstract : Boolean; 33 
  reference supertypes [*] : Class ; 34 
  reference structuralFeatures [*] ordered container : StructuralFeature 35 
oppositeOf owner ; 36 
  reference operations [*] ordered container : Operation oppositeOf owner ; 37 
 } 38 
 39 
 class TypedElement extends ModelElement { 40 
  attribute lower : Integer; 41 
  attribute upper : Integer; 42 
  attribute isOrdered : Boolean; 43 
  attribute isUnique : Boolean; 44 
  reference type : Classifier ; 45 
 } 46 
 47 
 class StructuralFeature extends TypedElement { 48 
  reference owner : Class oppositeOf structuralFeatures ; 49 
  reference subsetOf [*] : StructuralFeature oppositeOf derivedFrom ; 50 
  reference derivedFrom [*] : StructuralFeature oppositeOf subsetOf ; 51 
 } 52 
 53 
 class Attribute extends StructuralFeature { 54 
 } 55 
 56 
 class Reference extends StructuralFeature { 57 
  attribute isContainer : Boolean; 58 
  reference opposite [0- 1] : Reference ; 59 



ATL Transformation Example 
 

 
KM3 to Metrics Date 22/09/2005 

 

Page 7 

 } 60 
 61 
 class Operation extends TypedElement { 62 
  reference owner : Class oppositeOf operations ; 63 
  reference parameters [*] ordered container : Parameter oppositeOf owner ; 64 
 } 65 
 66 
 class Parameter extends TypedElement { 67 
  reference owner : Operation oppositeOf parameters ; 68 
 } 69 
 70 
 class Package extends ModelElement { 71 
  reference contents [*] ordered container : ModelElement oppositeOf "package" ; 72 
  reference metamodel : Metamodel oppositeOf contents ; 73 
 } 74 
 75 
 class Metamodel extends LocatedElement { 76 
  reference contents [*] ordered container : Package oppositeOf metamodel ; 77 
 } 78 
} 79 
 80 
package PrimitiveTypes { 81 
 datatype Boolean; 82 
 datatype Integer; 83 
 datatype String; 84 
} 85 

Appendix B: The Metrics metamodel in KM3 
format 

package Metrics { 1 
 2 
 abstract class Metric { 3 
  attribute context : String; 4 
  attribute label : String; 5 
 } 6 
  7 
 class StringMetric extends Metric { 8 
  attribute data : String; 9 
 } 10 
  11 
 class BooleanMetric extends Metric { 12 
  attribute data : Boolean; 13 
 } 14 
 15 
 class IntegerMetric extends Metric { 16 
  attribute data : Integer; 17 
 } 18 
} 19 
 20 
package PrimitiveTypes { 21 
 datatype Boolean; 22 
 datatype Integer; 23 
 datatype String; 24 
} 25 

Appendix C: The KM3 to Metrics ATL code 

module KM32Metrics; 1 
create OUT : Metrics from IN : KM3; 2 
 3 
--------------------------------------------------- ----------------------------  4 
-- HELPERS ---------------------------------------- ----------------------------  5 
--------------------------------------------------- ----------------------------  6 
 7 
-- This helper computes the set of all the Class el ements of the input model.  8 
-- CONTEXT: thisModule  9 
-- RETURN: Set(KM3!Class)  10 
helper def: allClasses : Set (KM3!Class) = KM3!Class.allInstances(); 11 
 12 
-- This helper computes the set of all the Referenc e elements of the input  13 
-- model.  14 
-- CONTEXT: thisModule  15 
-- RETURN: Set(KM3!Reference)  16 
helper def: allReferences : Set (KM3!Reference) = KM3!Reference.allInstances(); 17 
 18 
-- This helper computes the set of all the Class el ements of the input model  19 
-- that correspond to the root of the different inh eritance trees.  20 
-- CONTEXT: thisModule  21 
-- RETURN: Set(KM3!Class)  22 
helper def: inheritanceRoots : Set (KM3!Class) = 23 
 thisModule.allClasses 24 
  ->select(e | e.supertypes->isEmpty()); 25 
 26 
-- This helper returns the number of own attributes  of the contextual Class.  27 
-- CONTEXT: KM3!Class  28 
-- RETURN: Integer  29 
helper context KM3!Class def: attributeNb : Integer  = 30 
 self.structuralFeatures 31 
  ->select(e | e.oclIsTypeOf(KM3!Attribute)) 32 
  ->size(); 33 
 34 
-- This helper returns the total number of attribut es (including inherited  35 
-- ones) of the contextual Class.  36 
-- CONTEXT: KM3!Class  37 
-- RETURN: Integer  38 
helper context KM3!Class def: attributeNb2 : Integer  = 39 
 self.attributeNb +  40 
 self.supertypes 41 
  ->iterate(e; sum : Integer  = 0 | 42 
   sum + e.attributeNb 43 
  ); 44 
 45 
-- This helper returns the number of own references  of the contextual Class.  46 
-- CONTEXT: KM3!Class  47 
-- RETURN: Integer  48 
helper context KM3!Class def: referenceNb : Integer  = 49 
 self.structuralFeatures 50 
  ->select(e | e.oclIsTypeOf(KM3!Reference)) 51 
  ->size(); 52 
 53 
-- This helper returns the total number of referenc es (including inherited  54 
-- ones) of the contextual Class.  55 
-- CONTEXT: KM3!Class  56 
-- RETURN: Integer  57 
helper context KM3!Class def: referenceNb2 : Integer  = 58 
 self.referenceNb +  59 
 self.supertypes 60 
  ->iterate(e; sum : Integer  = 0 | 61 
   sum + e.referenceNb 62 
  ); 63 

 64 
-- This helper returns the size of the maximum leng ht of the inheritance trees  65 
-- that lead to the contextual Class.  66 
-- If the class has no supertype, the helper return s 0. Otherwise, it iterates  67 
-- through the set of supertypes of the contextual Class in order to find out  68 
-- the maximum lenght of inheritance trees (from ro ots to the contextual  69 
-- Class).  70 
-- CONTEXT: KM3!Class  71 
-- RETURN: Integer  72 
helper context KM3!Class def: getInheritanceLenght() : Integer  = 73 
 if self.supertypes->size() = 0 74 
 then 75 
  0 76 
 else 77 
  self.supertypes->iterate(e; max : Integer  = 0 | 78 
   if e.getInheritanceLenght() + 1 > max 79 
   then 80 
    e.getInheritanceLenght() + 1 81 
   else 82 
    max 83 
   endif 84 
  ) 85 
 endif; 86 
 87 
-- This helper returns the maximal depth of an inhe ritance tree in the input  88 
-- metamodel. For this purpose, it calls the getInh eritanceLenght() helper on  89 
-- each Class of the input model and returns the ma ximum value.  90 
-- CONTEXT: thisModule  91 
-- RETURN: Integer  92 
helper def: getInheritanceMaxDepth() : Integer  = 93 
 thisModule.allClasses 94 
  ->iterate(e; max : Integer  = 0 | 95 
   if e.getInheritanceLenght() > max 96 
   then 97 
    e.getInheritanceLenght() 98 
   else 99 
    max 100 
   endif 101 
  ); 102 
 103 
-- This helper returns the number of inheritance tr ees in the input model.  104 
-- For this purpose, it computes the number of "roo t" classes, that is the  105 
-- classes that donot have any supertype.  106 
-- CONTEXT: thisModule  107 
-- RETURN: Integer  108 
helper def: getInheritanceTreeNb() : Integer  = 109 
 thisModule.inheritanceRoots->size(); 110 
 111 
-- This helper returns the set of subtypes of its c ontextual Class.  112 
-- For this purpose, it selects among all existing Class elements, those that  113 
-- have the contextual Class as a supertype.  114 
-- CONTEXT: KM3!Class  115 
-- RETURN: Set(KM3!Class)  116 
helper context KM3!Class def: subTypes : Set (KM3!Class) = 117 
 thisModule.allClasses 118 
  ->select(e | e.supertypes->includes(self)) 119 
  ->asSet(); 120 
 121 
-- This helper computes the inheritance subtree of the contextual Class.  122 
-- For this purpose, the helper recursively calls i tself for each subtype of  123 
-- the contextual Class, adding this contextual Cla ss to the computed result.  124 
-- CONTEXT: KM3!Class  125 
-- RETURN: Set(KM3!Class)  126 
helper context KM3!Class def: getTree() : Set (KM3!Class) = 127 
 self.subTypes->iterate(e; tree : Set (KM3!Class) = Set {self} | 128 
  tree->union( e.getTree() ) 129 
 ); 130 
 131 
-- This helper computes a boolean value stating whe ther its exists an (or more)  132 

-- inheritance relation between classes of trees t1  and t2.  133 
-- The helper checks whether such a relation (super type/subtype) exists between  134 
-- each class of the tree t1 and the classes of the  tree t2.  135 
-- Comment sur les asSet()...  136 
-- CONTEXT: thisModule  137 
-- IN:  Set(KM3!Class), Set(KM3!Class)  138 
-- RETURN: Boolean  139 
helper def: existsHLink(t1 : Set (KM3!Class), t2 : Set (KM3!Class)) : Boolean  = 140 
 t1->iterate(e; res : Boolean  = false  | 141 
  if (e.supertypes->asSet()->intersection(t2)->isEmpty( ) and 142 
   e.subTypes->asSet()->intersection(t2)->isEmpty() ) 143 
  then 144 
   res 145 
  else 146 
   true  147 
  endif 148 
 ); 149 
 150 
-- This recursive helper computes the number of ind ependant inheritance graphs  151 
-- corresponding to the sequence of inheritance tre es passed as a parameter.  152 
-- If the input sequence contains a single tree, th e helper returns 1.  153 
-- Otherwise, the helper checks whether there exist s inheritance relationships  154 
-- (super/subtype) between classes of the first tre e of the input sequence (the  155 
-- reference tree), and the following ones. If so, it merges the linked trees  156 
-- into the reference tree.  157 
-- If the new tree sequence built this way contains  a single tree, the helper  158 
-- returns 1. Otherwise, the helper returns 1 + the  value provideed by a  159 
-- recursive call of itself on the newly calculated  tree sequence without its  160 
-- first reference tree.  161 
-- CONTEXT: thisModule  162 
-- IN:  Sequence(Set(KM3!Class))  163 
-- RETURN: Integer  164 
helper def: computeGraphNb(tree_seq : Sequence ( Set (KM3!Class))) : Integer  = 165 
 if tree_seq->size() = 1 166 
 then 167 
  1 168 
 else 169 
  let first_t : Set (KM3!Class) = tree_seq->first() in 170 
  let new_seq : Sequence ( Set (KM3!Class)) = 171 
  tree_seq 172 
   ->subSequence(2, tree_seq->size()) 173 
   ->iterate(e;t_seq : Sequence ( Set (KM3!Class)) = Sequence {first_t} | 174 
    if thisModule.existsHLink(first_t, e) 175 
    then 176 
     t_seq 177 
      ->subSequence(2, t_seq->size()) 178 
      ->prepend(t_seq->first()->including(e)) 179 
    else 180 
     t_seq.append(e) 181 
    endif 182 
   ) 183 
  in 184 
  if new_seq->size() = 1 185 
  then 186 
   1 187 
  else 188 
   thisModule.computeGraphNb( 189 
    new_seq->subSequence(2, new_seq->size()) 190 
   ) + 1 191 
  endif 192 
 endif; 193 
 194 
-- This helper returns the number of inheritage gra phs in the input model.  195 
-- For this purpose, it first computes a sequence c ontaining the set of classes  196 
-- representing the different inheritance trees of the input model. The helper  197 
-- then calls the recursive computeGraphNb with the  calculated sequence as a  198 
-- parameter.  199 
-- CONTEXT: thisModule  200 
-- RETURN: Integer  201 

helper def: getInheritanceGraphNb() : Integer  = 202 
 let tree_seq : Sequence ( Set (KM3!Class)) = 203 
  thisModule.inheritanceRoots 204 
   ->collect(e | e.getTree()) 205 
   ->asSequence() 206 
 in thisModule.computeGraphNb(tree_seq); 207 
 208 
 209 
--------------------------------------------------- ----------------------------  210 
-- RULES ------------------------------------------ ----------------------------  211 
--------------------------------------------------- ----------------------------  212 
 213 
-- Rule 'Model'  214 
-- This rule generates metrics elements that are re lative to the input model:  215 
-- * the number of classes of the model;  216 
-- * the number of associations of the model (this number corresponds to the  217 
--   number of references without an opposite, plus  the half of the number of  218 
--   references that has an opposite);  219 
-- * the maximal depth of an inheritance tree (comp uted by the  220 
--   getInheritanceMaxDepth helper);  221 
-- * the number of inheritance trees (computed by t he getInheritanceTreeNb  222 
--   helper);  223 
-- * the number of inheritance graphs (computed by the getInheritanceGraphNb  224 
--   helper).  225 
rule Model { 226 
 from 227 
  i : KM3!Metamodel 228 
 to 229 
  o1 : Metrics!IntegerMetric ( 230 
   " context" <- 'Model' , 231 
   label <- 'Number of classes' , 232 
   data <- thisModule.allClasses->size() 233 
  ), 234 
   235 
  o2 : Metrics!IntegerMetric ( 236 
   " context" <- 'Model' , 237 
   label <- 'Number of associations' , 238 
   data <- 239 
    (thisModule.allReferences 240 
     ->select(e | e.opposite = OclUndefined) 241 
     ->size()) 242 
    + 243 
    ((thisModule.allReferences 244 
     ->select(e | e.opposite <> OclUndefined) 245 
     ->size()) div 2) 246 
  ), 247 
   248 
  o3 : Metrics!IntegerMetric ( 249 
   " context" <- 'Model' , 250 
   label <- 'Maximal depth of heritance' , 251 
   data <- thisModule.getInheritanceMaxDepth() 252 
  ), 253 
   254 
  o4 : Metrics!IntegerMetric ( 255 
   " context" <- 'Model' , 256 
   label <- 'Number of inheritance trees' , 257 
   data <- thisModule.getInheritanceTreeNb() 258 
  ), 259 
   260 
  o5 : Metrics!IntegerMetric ( 261 
   " context" <- 'Model' , 262 
   label <- 'Number of inheritance graphs' , 263 
   data <- thisModule.getInheritanceGraphNb() 264 
  ) 265 
} 266 
 267 
-- Rule 'Class'  268 
-- This rule generates metrics elements taht are re lative to a Class of the  269 
-- input model. The context of the generated metric s elements therefore  270 



ATL Transformation Example 
 

 
KM3 to Metrics Date 22/09/2005 

 

Page 13 

-- includes the name of the class theyr refer to.  271 
-- Generated metrics are:  272 
-- * the number of own attributes of the class;  273 
-- * the number of own references of the class;  274 
-- * the total number of attributes of the class (i ncluding inherited ones);  275 
-- * the total number of references of the class (i ncluding inherited ones);  276 
rule Class { 277 
 from 278 
  i : KM3!Class 279 
 to 280 
  o1 : Metrics!IntegerMetric ( 281 
   " context" <- 'Class '  + i.name, 282 
   label <- 'Number of own attributes' , 283 
   data <- i.attributeNb 284 
  ), 285 
 286 
  o2 : Metrics!IntegerMetric ( 287 
   " context" <- 'Class '  + i.name, 288 
   label <- 'Number of own references' , 289 
   data <- i.referenceNb 290 
  ), 291 
   292 
  o3 : Metrics!IntegerMetric ( 293 
   " context" <- 'Class '  + i.name, 294 
   label <- 'Total number of attributes' , 295 
   data <- i.attributeNb2 296 
  ), 297 
 298 
  o4 : Metrics!IntegerMetric ( 299 
   " context" <- 'Class '  + i.name, 300 
   label <- 'Total number of references' , 301 
   data <- i.referenceNb2 302 
  ) 303 
} 304 


